Introduction to Bayesian Statistics Project: Bayesian Inference in Finance: Volatility Modeling

Chorobek Sheranov

Central European University

11.12.2023

Introduction to Volatility

- Volatility measures the **degree of variation** in the price of financial instruments over time.
- It reflects the **pace and magnitude** of price fluctuations.
- Volatility is often represented in statistical terms, such as the **standard deviation** of price movements.
- Statistical notation for volatility provides a quantitative measure of the dispersion of asset prices.



Why Volatility Matters

- **Risk Assessment:** Volatility helps investors gauge the potential risk associated with a financial asset.
- **Market Conditions:** High volatility may indicate uncertain market conditions, while low volatility suggests stability.
- **Volatility Trading:** Exploit discrepancies between forecasted volatility and implied volatility. If forecasted volatility is significantly different from implied volatility, it may create arbitrage opportunities.

Option Pricing and the Black-Scholes Model:

- In options trading, volatility is a key input in pricing models like the **Black-Scholes Model**.
- The Black-Scholes Model helps estimate the fair market value of options, considering factors like volatility, time to expiration, and underlying asset price.

Exploring Volatility Modeling

Volatility Modeling:

- Volatility modeling is crucial for assessing and managing financial risk.
- In this exploration, we focus on two approaches: GARCH(1,1) and Stochastic Volatility Models.

Models Under Consideration:

1. **GARCH(1,1):** A popular time series model capturing volatility dynamics.
2. **Stochastic Volatility Model:** Incorporates latent stochastic processes for flexible volatility modeling.

Objective:

- Evaluate the effectiveness of GARCH(1,1) and Stochastic Volatility Models in capturing and predicting financial volatility.

Generalized Autoregressive Conditional Heteroscedasticity (GARCH) Model

- Introduced by Robert Engle in 1982.
- Extension of the ARCH model (1982), designed to capture time-varying volatility in financial time series.

GARCH(1,1) Model Formulation:

$$\sigma_t^2 = \alpha_0 + \alpha_1 a_{t-1}^2 + \beta_1 \sigma_{t-1}^2 \tag{1}$$

- σ_t^2: Conditional variance at time t.
- a_{t-1}^2: Squared previous difference in return from the mean at time $t-1$.
- $\alpha_0, \alpha_1, \beta_1$: Model coefficients.

Historical Context:

- Developed to address the limitations of constant volatility assumptions in financial modeling.
- Widely used in modeling and forecasting financial market volatility.

GARCH(1, 1)

GARCH(1,1) Model Formulation:

$$\sigma_t^2 = \alpha_0 + \alpha_1 a_{t-1}^2 + \beta_1 \sigma_{t-1}^2$$

- σ_t^2: Conditional variance at time t.
- a_{t-1}^2: Squared previous difference in return from the mean at time $t-1$.
- $\alpha_0, \alpha_1, \beta_1$: Model coefficients.

Model Constraints:

- To ensure positivity and stationarity, coefficients must satisfy $\alpha_0, \alpha_1, \beta_1 > 0$.
- The sum of the slopes $\alpha_1 + \beta_1$ must be less than 1.

Note that the conditional variance σ_t^2 is a deterministic function of the model parameters and past data

GARCH Models: Assumptions and Suitability

Key Assumptions of GARCH Models:

- **Conditional Heteroskedasticity:** Assumes varying conditional variance based on past observations.
- **Stationarity:** Assumes covariance-stationary time series.
- **Normality of Residuals:** Assumes normality of squared return residuals.

Best for Modeling:

- **Highly Oscillated Data:** Effective for volatility clustering in financial time series.
- **Stable and Unstable Periods:** Versatile across various market conditions.

Summary: GARCH models excel in modeling time-varying volatility, especially during volatility clustering. Versatile across market conditions, but caution needed with non-stationary data and extreme events.

Stochastic Volatility Models: Introduction

Stochastic Volatility Models (SVM):

- Treat volatility (variance) of an asset's return as a latent stochastic process.
- Developed by Kim, Shephard, and Chib in 1998.

Simplest Stochastic Volatility Model:

- Introduces a latent process governing volatility evolution over time.
- Captures time-varying volatility more flexibly than GARCH.

Difference from GARCH:

- GARCH models assume constant volatility with discrete changes.
- Stochastic Volatility models allow for continuous, smooth changes in volatility.

Stochastic Volatility Model: Equations & Notations

$$y_t = \epsilon_t \exp\left(\frac{h_t}{2}\right)$$
$$h_{t+1} = \mu + \phi(h_t - \mu) + \delta_t \sigma$$
$$h_1 \sim \mathcal{N}\left(\mu, \frac{\sigma}{\sqrt{1-\phi^2}}\right)$$
$$\epsilon_t \sim \mathcal{N}(0,1)$$
$$\delta_t \sim \mathcal{N}(0,1)$$

Sampling Distributions:

- $y_t \sim \mathcal{N}(0, \exp(h_t/2))$
- $h_t \sim \mathcal{N}(\mu + \phi(h_{t-1} - \mu), \sigma)$

Notations:

y_t: Mean-corrected returns on the underlying asset

h_t: Log volatility at time t

μ: Mean log volatility

ϕ: Persistence of the volatility term

ϵ_t: White-noise shock on asset return at time t

δ_t: Shock on volatility at time t

Role of Bayesian Inference in GARCH and SVM

Bayesian Inference:

- **Optimizing Parameters:** Maximum likelihood estimation is commonly used for parameter optimization.
- **Potential Pitfall:** Optimum parameters based solely on training data may lead to suboptimal models if the data is atypical or not the best representative of the underlying process.

Bayesian Approach:

- **Posterior Distribution:** Bayesian Inference computes the posterior distribution of parameters.
- **Robust Estimates:** Sampling from the posterior allows for deeper exploration, leading to more robust estimates.
- **Power of Priors:** Properly chosen priors enhance model generalization for long-lasting performance.

Practical Assessment: Modeling the Weekly Volatility of S&P 500 using Bayesian Approach with GARCH(1,1) and Stochastic Volatility Models

Stan code for GARCH(1, 1)

```
data {
      int<lower=0> T;
      array[T] real r;
      real<lower=0> sigma1;
    }
    parameters {
      real mu;
      real<lower=0> alpha0;
      real<lower=0, upper=1> alpha1;
      real<lower=0, upper=(1-alpha1)> beta1;
    }
    transformed parameters {
      array[T] real<lower=0> sigma;
      sigma[1] = sigma1;
      for (t in 2:T) {
        sigma[t] = sqrt(alpha0
                        + alpha1 * pow(r[t - 1] - mu, 2)
                        + beta1 * pow(sigma[t - 1], 2));
      }
    }
    model {
      // Priors
      mu ~ normal(0, 10);          // Prior for mean parameter
      alpha0 ~ cauchy(0, 5);       // Prior for alpha0
      alpha1 ~ beta(2, 2);          // Prior for alpha1 (beta distribution ensures [0, 1])
      beta1 ~ uniform(0, 1);        // Prior for beta1 (uniform distribution ensures [0, 1])

      r ~ normal(mu, sigma);
    }

    generated quantities {   //log likelihood for loo-cv
      vector[T] log_lik;

      for (t in 1:T) {
        log_lik[t] = normal_lpdf(r[t] | mu, sigma[t]);
      }
```

Diagnostics for GARCH(1, 1)

	mean	sd	hdi_3%	hdi_97%	mcse_mean	mcse_sd	ess_bulk	ess_tail	r_hat
mu	0.003	0.001	0.002	0.005	0.000	0.0	44505.0	29643.0	1.0
alpha0	0.000	0.000	0.000	0.000	0.000	0.0	12397.0	14215.0	1.0
alpha1	0.154	0.051	0.064	0.249	0.000	0.0	15595.0	16883.0	1.0
beta1	0.778	0.069	0.653	0.899	0.001	0.0	13391.0	14543.0	1.0



Diagnostics for GARCH(1, 1)



Diagnostics for GARCH(1, 1)



Stochastic Volatility Model: Stan Code

```
data {
  int<lower=0> T;   // # time points (equally spaced)
  vector[T] y;      // mean corrected return at time t
}
parameters {
  real mu;                     // mean log volatility
  real<lower=-1, upper=1> phi; // persistence of volatility
  real<lower=0> sigma;         // white noise shock scale
  vector[T] h_std;             // standardized log volatility at time t
}
transformed parameters {
  vector[T] h = h_std * sigma;  // now h ~ normal(0, sigma)
  h[1] /= sqrt(1 - phi * phi);  // rescale h[1]
  h += mu;
  for (t in 2:T) {
    h[t] += phi * (h[t - 1] - mu);
  }
}
model {
  phi ~ normal(0.885, 0.5);    //priors
  sigma ~ cauchy(0, 5);
  mu ~ normal(-8.1, 2);
  h_std ~ std_normal();  // vectorized standard normal sampling
  y ~ normal(0, exp(h / 2));
}
generated quantities {
  vector[T] log_lik;

  for (t in 1:T) {
    // Add a small constant to the denominator to avoid zero scale or to regularize
    real scale = exp(h[t] / 2) + 1e-6;
    log_lik[t] = normal_lpdf(y[t] | 0, scale);
  }
}
```

Diagnostics for the SV Model

	mean	sd	hdi_3%	hdi_97%	mcse_mean	mcse_sd	ess_bulk	ess_tail	r_hat
mu	-8.809	0.431	-9.692	-8.038	0.011	0.008	3224.0	1371.0	1.0
phi	0.932	0.062	0.825	0.998	0.001	0.000	4503.0	1761.0	1.0
sigma	0.277	0.112	0.096	0.486	0.001	0.001	8865.0	13497.0	1.0



Diagnostics for the SV Model



Diagnostics for the SV Model



As we can see, posterior distributions of parameters for the stochastic volatility are much more complex, therefore require more iterations and other adjustments such as using informative priors for obtaining valid estimates and good r-hat scores.

Posterior Predictive Checking: Garch(1, 1)



Posterior Predictive Checking: SV Model



Predictive Performance Assessment: Garch(1, 1)



Predictive Performance Assessment: SV Model



Model Comparison: MSE



Model Comparison: LOO-CV

	rank	elpd_loo	p_loo	elpd_diff	weight	se	dse	warning	scale
SV	0	745.612191	21.643159	0.000000	0.998913	13.557279	0.000000	True	log
GARCH	1	735.597810	4.288769	10.014381	0.001087	15.217783	5.388008	False	log



Figure: Pareto GA

Figure: Pareto SV

Conclusions on the Models

Based on MSE scores and leave-one-out cross-validation model comparison, the stochastic volatility model appears to outperform the GARCH(1,1) model. This result is consistent with the general observation that stochastic volatility models tend to perform better, especially in longer-period estimations.

An interesting observation is that the GARCH model accurately predicted the spike in volatility during the COVID-19 period, while the stochastic volatility model struggled in this specific task. However, during more stable intervals, the stochastic volatility model demonstrated higher accuracy.

Conclusions on the Models

In summary, both models are doing fairly well, especially accounting for the fact that they are the most simplistic ones. There are many variations such as with moving averages, jump factors, etc. which have more parameters and can give more sophisticated results particularly for the estimation of the covid spike.

There is no such a thing as a bad model, each model has its own task and own environment it performs best in. So based on the nature of the problem, corresponding choice of models to train is done.

Key Insights and Conclusions

Versatile Applications of Bayesian Methods

Bayesian methodologies showcase their versatility across diverse scientific domains. In our exploration today, we have specifically focused on their applications in the intricate landscape of finance.

Navigating Data Uncertainties

In an age where data abundance is the norm, there are instances where the available data might be limited, or its representativeness in capturing the essence of a complex problem remains uncertain. Bayesian methods emerge as a robust solution to address such uncertainties.

Bayesian Framework: Incorporating Prior Beliefs

The Bayesian framework allows us to seamlessly integrate prior beliefs into our analysis. This unique feature becomes particularly crucial when dealing with scenarios where the available data alone might not provide a comprehensive representation of the underlying dynamics.